NAG Fortran Library Routine Document

G05PAF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05PAF generates a realisation of a univariate time series from an autoregressive moving average (ARMA) model. The realisation may be continued or a new realisation generated at subsequent calls to G05PAF.

2 Specification

```
SUBROUTINE G05PAF(MODE, XMEAN, IP, PHI, IQ, THETA, AVAR, VAR, N, X,
1                 IGEN, ISEED, R, NR, IFAIL)
INTEGER           MODE, IP, IQ, N, IGEN, ISEED(4), NR, IFAIL
real              XMEAN, PHI(*), THETA(*), AVAR, VAR, X(*), R(NR)
```

3 Description

Let the vector x_t, denote a time series which is assumed to follow an autoregressive moving average (ARMA) model of the form:

$$x_t - \mu = \phi_1(x_{t-1} - \mu) + \phi_2(x_{t-2} - \mu) + \ldots + \phi_p(x_{t-p} - \mu) + \epsilon_t - \theta_1\epsilon_{t-1} - \theta_2\epsilon_{t-2} - \ldots - \theta_q\epsilon_{t-q} \quad (1)$$

where ϵ_t, is a residual series of independent random perturbations assumed to be Normally distributed with zero mean and variance σ^2. The parameters $\{\phi_i\}$, for $i = 1, 2, \ldots, p$, are called the autoregressive (AR) parameters, and $\{\theta_j\}$, for $j = 1, 2, \ldots, q$, the moving average (MA) parameters. The parameters in the model are thus the $p\phi$-values, the $q\theta$-values, the mean μ and the residual variance σ^2.

The routine sets up a reference vector containing initial values corresponding to a stationary position using the method described in Tunnicliffe-Wilson (1979). The routine can then return a realisation of $x_1, x_2, \ldots, x_n$. On a successful exit, the recent history is updated and saved in the reference vector R so that G05PAF may be called again to generate a realisation of $x_{n+1}, x_{n+2}, \ldots$,etc. See the description of the parameter MODE in Section 5 for details.

One of the initialisation routines G05KBF (for a repeatable sequence if computed sequentially) or G05KCF (for a non-repeatable sequence) must be called prior to the first call to G05PAF.

4 References

Knuth D E (1981) *The Art of Computer Programming (Volume 2)* (2nd Edition) Addison-Wesley

Tunnicliffe-Wilson G (1979) Some efficient computational procedures for high order ARMA models *J. Statist. Comput. Simulation* **8** 301–309

5 Parameters

1: MODE – INTEGER *Input*

On entry: a code for selecting the operation to be performed by the routine:

MODE = 0

Set up reference vector only.

MODE = 1

Generate terms in the time series using reference vector set up in a prior call to G05PAF.

MODE = 2

Set up reference vector and generate terms in the time series.

Constraint: MODE = 0, 1 or 2.

2: XMEAN – ***real*** *Input*

On entry: the mean of the time series.

3: IP – INTEGER *Input*

On entry: the number of autoregressive coefficients supplied, p.

Constraint: IP ≥ 0.

4: PHI(∗) – ***real*** array *Input*

Note: the dimension of the array PHI must be at least $\max(1, \text{IP})$.

On entry: the autoregressive coefficients of the model, $\phi_1, \phi_2, \ldots, \phi_p$.

5: IQ – INTEGER *Input*

On entry: the number of moving average coefficients supplied, q.

Constraint: IQ ≥ 0.

6: THETA(∗) – ***real*** array *Input*

Note: the dimension of the array THETA must be at least $\max(1, \text{IQ})$.

On entry: the moving average coefficients of the model, $\theta_1, \theta_2, \ldots, \theta_q$.

7: AVAR – ***real*** *Input*

On entry: the variance of the normal perturbations, σ^2.

8: VAR – ***real*** *Output*

On exit: the proportion of the variance of a term in the series that is due to the moving-average (error) terms in the model. The smaller this is, the nearer is the model to non-stationarity.

9: N – INTEGER *Input*

On entry: the number of observations to be generated, n.

Constraint: N ≥ 0.

10: X(∗) – ***real*** array *Output*

Note: the dimension of the array X must be at least $\max(1, \text{N})$.

On exit: contains the next n observations from the time series.

11: IGEN – INTEGER *Input*

On entry: must contain the identification number for the generator to be used to return a pseudo-random number and should remain unchanged following initialisation by a prior call to one of the routines G05KBF or G05KCF.

12: ISEED(4) – INTEGER array *Input/Output*

On entry: contains values which define the current state of the selected generator.

On exit: contains updated values defining the new state of the selected generator.

13: R(NR) – ***real*** array *Input/Output*

On exit: the reference vector.

14: NR – INTEGER *Input*

On entry: the dimension of the array R as declared in the (sub)program from which G05PAF is called.

Constraint: NR $\geq$ IP + IQ + 5 + max(IP, IQ + 1).

15: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, for users not familiar with this parameter the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, N $<$ 0.

IFAIL = 2

On entry, IP $<$ 0.

IFAIL = 3

On entry, IQ $<$ 0.

IFAIL = 4

On entry, AVAR $<$ 0.

IFAIL = 5

On entry, MODE $<$ 0,
or MODE $>$ 2.

IFAIL = 6

On entry, NR $<$ IP + IQ + 5 + max(IP, IQ + 1).

IFAIL = 7

PHI does not define a stationary autoregresive process.

IFAIL = 8

Either R has been corrupted or the value of IP or IQ is not the same as when R was set up in a previous call with MODE = 0 or 2.

7 Accuracy

The errors in the initialisation process should be very much smaller than the error term; see Tunnicliffe-Wilson (1979).

8 Further Comments

The time taken by the routine is essentially of order $(\mathrm{IP})^2$.

Note: G05KBF and G05KCF must be used with care if this routine is used as well. The reference vector, as mentioned before, contains a copy of the recent history of the series. This will not be altered properly by calls to any of the above routines. A call to G05KBF or G05KCF should be followed by a call to G05PAF with MODE = 0 to re-initialise the time series reference vector in use. To maintain repeatability with G05KBF, the calls to G05PAF should be performed in the same order and at the same point or points in the simulation every time G05KBF is used. When the generator state is saved and restored using the parameters IGEN and ISEED, the time series reference vector must be saved and restored as well.

The ARMA model for a time series can also be written as:

$$(x_n - E) = A_1(x_{n-1} - E) + \ldots + A_{NA}(x_{n-NA} - E) + B_1 a_n + \ldots + B_{NB} a_{n-NB+1}$$

where

x_n is the observed value of the time series at time n,

NA is the number of autoregressive parameters, A_i,

NB is the number of moving average parameters, B_i,

E is the mean of the time series,

and

a_t is a series of independent random Standard Normal perturbations.

This is the form used in G05EGF. This is related to the form given in Section 3 by:

$B_1^2 = \sigma^2$,

$B_{i+1} = -\theta_i \sigma = -\theta_i B_1, \quad i = 1, 2, \ldots, q$,

$\mathrm{NB} = q + 1$,

$E = c$,

$A_i = \phi_i, \quad i = 1, 2, \ldots, p$,

$\mathrm{NA} = p$.

9 Example

This example program calls G05PAF to set up the reference vector for an autoregressive model after initialisation by G05KBF. The model is given by

$$x_t = 0.4x_{t-1} + 0.2x_{t-2} + \epsilon_t$$

where ϵ_t is a series of independent random Normal perturbations with variance 1.0. G05PAF is then called generate a sample of ten observations, which are printed.

9.1 Program Text

Note: the listing of the example program presented below uses ***bold italicised*** terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     G05PAF Example Program Text
*     Mark 20 Release. NAG Copyright 2001.
*     .. Parameters ..
      INTEGER          IP, IQ, N, NR
      PARAMETER        (IP=2,IQ=0,N=10,NR=IP+IQ+5+IP)
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
*     .. Local Scalars ..
      real             AVAR, VAR, XMEAN
      INTEGER          I, IFAIL, IGEN
*     .. Local Arrays ..
      real             PHI(IP), R(NR), THETA(1), X(N)
      INTEGER          ISEED(4)
*     .. External Subroutines ..
      EXTERNAL         G05KBF, G05PAF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G05PAF Example Program Results'
      WRITE (NOUT,*)
*     Set the ARMA model parameters
      XMEAN = 0.0e0
      PHI(1) = 0.4e0
      PHI(2) = 0.2e0
      AVAR = 1.0e0
*     Initialise the seed to a repeatable sequence
      ISEED(1) = 1762543
      ISEED(2) = 9324783
      ISEED(3) = 4234401
      ISEED(4) = 742355
*     IGEN identifies the stream.
      IGEN = 1
      CALL G05KBF(IGEN,ISEED)
      IFAIL = 0
*     Set up the reference vector
      CALL G05PAF(0,XMEAN,IP,PHI,IQ,THETA,AVAR,VAR,N,X,IGEN,ISEED,R,NR,
     +            IFAIL)
*     Generate a sample of 10 observations
      CALL G05PAF(1,XMEAN,IP,PHI,IQ,THETA,AVAR,VAR,N,X,IGEN,ISEED,R,NR,
     +            IFAIL)
      DO 20 I = 1, N
         WRITE (NOUT,99999) X(I)
   20 CONTINUE
      STOP
*
99999 FORMAT (1X,F12.4)
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 G05PAF Example Program Results

      -1.0654
      -0.2828
      -2.0924
      -2.3304
      -2.5998
      -1.7143
      -2.4882
      -1.3882
      -2.2722
      -1.8806
```